[1] DALINIŲ IŠVESTINIŲ LYGTIS (20 VARIANTAS)

Suvedame lygtį į kanoninį pavidalą ir randame bendrąjį sprendinį.

```
load(pdiff)$
```

Kintamųjų keitimui apibrėžiame funkciją "changevars"

```
changevars(eq,tr,itr):=block([a,b,c,d,eq1],
[a,b,c]:[coeff(lhs(eq),'diff(u,x,2)),coeff(lhs(eq),'diff(u,x,1,y,1))/2,coeff(lhs(eq),'diff(u,y,2))],
d:b^2−a·c,
depends(u,[ξ,η],ξ,[x,y],η,[x,y]),
append(diff(tr,x),diff(tr,x,2),diff(tr,y),diff(tr,y,2),diff(tr,x,1,y,1)),
subst(%%,ev(eq,nouns)),
subst(itr,%%),
ratsimp(%%),
trigsimp(%%),
expand(%%),
eq1:trigreduce(%%),
if d<0 then
(solve(eq1,diff(u,ξ,2))[1],
factor(%%+diff(u,η,2)))
elseif d=0 then factor(eq1)
else
(solve(eq1,diff(u,ξ,1,η,1))[1],
factor(%%)),
lhs(%%)−rhs(%%)=0
)$
```

```
eqtype(eq):=block([a,b,c,d],
a:coeff(lhs(eq),'diff(u,x,2)),
b:coeff(lhs(eq),'diff(u,x,1,y,1))/2,
c:coeff(lhs(eq),'diff(u,y,2)),
d:b^2−a·c,
if d>0 then "equation is hyperbolic"
elseif d=0 then "equation is parabolic"
elseif d<0 then "equation is elliptic"
)$
```

```
eq:1·'diff(u,x,2)+32·'diff(u,x,1,y,1)+192·'diff(u,y,2)=0;
```

$$192\left(\frac{d^2}{d\,y^2}\,u\right)+\frac{d^2}{d\,x^2}\,u+32\left(\frac{d^2}{d\,x\,d\,y}\,u\right)=0$$

```
[a,b,c]:[coeff(lhs(eq),'diff(u,x,2)),coeff(lhs(eq),'diff(u,x,1,y,1))/2,coeff(lhs(eq),'diff(u,y,2))];
```

$[1,16,192]$

```
d:b^2−a·c;
```

64

```
eqtype(eq);
```

equation is hyperbolic

```
cheq1:'diff(y,x)=(b+sqrt(b^2−a·c))/a;
```

$$\frac{d}{d\,x}\,y=24$$

```
ode2(cheq1,y,x);
```

$$y=24\,x+\%c$$

```
r1:solve(%,%c)[1];
```

$$\%c=y-24\,x$$

```
cheq2:'diff(y,x)=(b−sqrt(b^2−a·c))/a;
```

$$\frac{d}{d\,x}\,y=8$$

```
ode2(cheq2,y,x);
```

$$y=8\,x+\%c$$

```
r2:solve(%,%c)[1];
```

$$\%c=y-8\,x$$

Gavome transformacijos formules:

```
tr:[ξ=rhs(r1),η=rhs(r2)];
```

$$[\xi=y-24\,x,\eta=y-8\,x]$$

```
itr:solve(tr,[x,y])[1];
```

$$\left[x=-\frac{\xi-\eta}{16},y=-\frac{\xi-3\,\eta}{2}\right]$$

```
eq1:changevars(eq,tr,itr);
```

$$\frac{d^2}{d\,\eta\,d\,\xi}\,u=0$$

```
canonical_form:%;
```

$$\frac{d^2}{d\,\eta\,d\,\xi}\,u=0$$

Bendrasis sprendinys:

```
u=C1(ξ)+C2(η);
```

$$u=\mathrm{C1}\left(\xi\right)+\mathrm{C2}\left(\eta\right)$$

Atsakymas:

```
ats:subst(tr,%);
```

$$u=\mathrm{C2}\left(y-8\,x\right)+\mathrm{C1}\left(y-24\,x\right)$$

Patikrinimas:

```
subst(ats,eq);
```

$$192\left(\frac{d^2}{d\,y^2}\left(C2\left(y-8\,x\right)+C1\left(y-24\,x\right)\right)\right)+\frac{d^2}{d\,x^2}\left(C2\left(y-8\,x\right)+C1\left(y-24\,x\right)\right)+32\left(\frac{d^2}{d\,x\,d\,y}\left(C2\left(y-8\,x\right)+C1\left(y-24\,x\right)\right)\right)=0$$

```
ev(%,diff);
```

$$192\left(C2_2\left(y-8\,x\right)+C1_2\left(y-24\,x\right)\right)+64\,C2_2\left(y-8\,x\right)+32\left(-8\,C2_2\left(y-8\,x\right)-24\,C1_2\left(y-24\,x\right)\right)+576\,C1_2\left(y-24\,x\right)=0$$

```
expand(%);
```

$$0=0$$

```
is(%);
```

true