

[1] Antros eilės dif. lygties sprendimas

20 variantas: $x'' + 9x = 2\sin(t) - 6\cos(2t)$, $x(0) = 0$, $x'(0) = 4$

`eq:'diff(x,t,2)+9*x=2*sin(t)-6*cos(2*t);`

$$\frac{d^2}{dt^2} x + 9x = 2 \sin(t) - 6 \cos(2t)$$

Naudojant ODE2 funkciją

`ode2(eq,x,t);`

$$x = \%k1 \sin(3t) + \%k2 \cos(3t) - \frac{24 \cos(2t) - 5 \sin(t)}{20}$$

Pertvarkome šį sprendinį

`expand(%);`

$$x = \%k1 \sin(3t) + \%k2 \cos(3t) - \frac{6 \cos(2t)}{5} + \frac{\sin(t)}{4}$$

Įvedame pradines sąlygas

`ic2(%, t=0, x=0, 'diff(x,t)=4);`

$$x = \frac{5 \sin(3t)}{4} + \frac{6 \cos(3t)}{5} - \frac{6 \cos(2t)}{5} + \frac{\sin(t)}{4}$$

`ats:rhs(%);`

$$\frac{5 \sin(3t)}{4} + \frac{6 \cos(3t)}{5} - \frac{6 \cos(2t)}{5} + \frac{\sin(t)}{4}$$

Patikrinimas:

`subst(x=ats,eq);`

$$\frac{d^2}{dt^2} \left[\frac{5 \sin(3t)}{4} + \frac{6 \cos(3t)}{5} - \frac{6 \cos(2t)}{5} + \frac{\sin(t)}{4} \right] + 9 \left[\frac{5 \sin(3t)}{4} + \frac{6 \cos(3t)}{5} - \frac{6 \cos(2t)}{5} + \frac{\sin(t)}{4} \right] = 2 \sin(t) - 6 \cos(2t)$$

`ev(%, diff);`

$$9 \left[\frac{5 \sin(3t)}{4} + \frac{6 \cos(3t)}{5} - \frac{6 \cos(2t)}{5} + \frac{\sin(t)}{4} \right] - \frac{45 \sin(3t)}{4} - \frac{54 \cos(3t)}{5} + \frac{24 \cos(2t)}{5} - \frac{\sin(t)}{4} = 2 \sin(t) - 6 \cos(2t)$$

```
expand(%);
```

$$2 \sin(t) - 6 \cos(2t) = 2 \sin(t) - 6 \cos(2t)$$

```
is(%);
```

true

```
subst(t=0,ats);
```


0

```
subst(t=0,diff(ats,t));
```


4

Sprendinio tyrimas:

```
wxplot2d([ats], [t,-10,10])$
```



```
wxplot2d(ats, [t,0,2*%pi], [y,-5,6])$
```



Ar sprendinys periodinis?

```
ats=subst(t=t+2*%pi,ats);
```

$$\begin{aligned} & - \frac{5 \sin(3(t+2\pi))}{4} - \frac{6 \cos(3(t+2\pi))}{5} + \frac{6 \cos(2(t+2\pi))}{5} + \frac{5 \sin(3t)}{4} + \\ & \frac{6 \cos(3t)}{5} - \frac{6 \cos(2t)}{5} \end{aligned}$$

```
trigexpand(%);
```

0

```
expand(%);
```

0

```
is(%=0);
```

true

Rasime sprendinio amplitudę, kai $0 < t < 2\pi$.

```
f1:=diff(ats,t); /* skaičiuojame išvestinę */
```

$$- \frac{18 \sin(3t)}{5} + \frac{15 \cos(3t)}{4} + \frac{12 \sin(2t)}{5} + \frac{\cos(t)}{4}$$

```
t1:=find_root(f1=0, t, 3, 4); /* ieškome "išvestinė=0" šaknies intervale [3,4]
(iš grafiko matome, kad šiame intervale yra
minimumo taškas) */
```

3.360430986952715

```
f_min:=subst(t=t1,ats); /* skaičiuojame funkcijos minimalią reikšmę */
```

-2.854659520680367

```
t2:=find_root(f1=0, t, 4, 5); /* ieškome "išvestinė=0" šaknies intervale [4,5]
(iš grafiko matome, kad šiame intervale yra
maksimumo taškas) */
```

4.513981216029051

```
f_max:=subst(t=t2,ats); /* skaičiuojame funkcijos maksimalią reikšmę */
```

2.569521822986567

```
Amplitude:=f_max-f_min; /* suskaičiuojame amplitudę */
```

5.424181343666934