

## [3] PIRMOS EILĖS NETIESINIŲ DIFERENCIALINIŲ LYGČIŲ SISTEMA

Lotkos-Volteros lygtys (plėšrūno-aukos lygtys):

$$\begin{aligned}x' &= (a - b \cdot y) \cdot x, \\ y' &= (-c + d \cdot x) \cdot y,\end{aligned}$$

$$\begin{aligned}x(0) &= x_0, \\ y(0) &= y_0,\end{aligned}$$

x - auka,  
y - plėšrūnas.

`load(drawdf)$`

`[a,b,c,d]:[4,3,4,3]$`

`eqs:[(a-b*y)*x,(-c+d*x)*y];`

$$\left[ x(4-3y), (3x-4)y \right]$$

Pusiausvyros taškai:

`solve(eqs,[x,y]);`

$$\left[ \left[ x=0, y=0 \right], \left[ x=\frac{4}{3}, y=\frac{4}{3} \right] \right]$$

Imame netrivialųjį pusiausvyros tašką:

`C:subst(%[2],[x,y]);`


$$\left[ \frac{4}{3}, \frac{4}{3} \right]$$

Fazinis portretas plėšrūnų-aukų įvairiomis pradinėmis sąlygomis  $x_0$  (pradinis aukų skaičius) ir  $y_0$  (pradinis plėšrūnų skaičius). Trajektorijos yra uždaros.

```

wxdrawdf(eqs, [x,0,5], [y,0,5],
line_width=2,
color=green, key = "x0=1, y0=2", soln_at(1,2),
color=blue, key = "x0=1, y0=3", soln_at(1,3),
color=purple, key = "x0=1, y0=4", soln_at(1,4),
point_type = filled_circle,
color=red,
xlabel="x (aukos)", ylabel="y (plėšrūnai)",
key = "Pusiausvyros taškas",
points([C])
)$

```




```
sol: rk(eqs,[x,y],[1,3],[t,0,8,0.02])$
```

Ištiesintas dif. lygčių sistemos sprendinys:

```

wxplot2d ([
  [discrete,makelist([p[1],p[2]],p,sol)],
  [discrete,makelist([p[1],p[3]],p,sol)]
],
[legend,"aukos", "plēšrūnai"],
[xlabel,"Laikas, t"],[ylabel,"Populiacija"]
)$

```



```

wxplot2d ([discrete,makelist([p[2],p[3]],p,sol)], [xlabel,"x"],[ylabel,"y"])$

```

